NAG Library Routine Document

F07WEF (DPFTRS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07WEF (DPFTRS) solves a real symmetric positive definite system of linear equations with multiple right-hand sides,

$$AX = B,$$

where A has been factorized by F07WDF (DPFTRF), stored in Rectangular Full Packed (RFP) format. The RFP storage format is described in Section 3.3.3 in the F07 Chapter Introduction.

2 Specification

```
SUBROUTINE F07WEF (TRANSR, UPLO, N, NRHS, A, B, LDB, INFO)

INTEGER             N, NRHS, LDB, INFO
REAL (KIND=nag_wp) A(N*(N+1)/2), B(LDB,*)
CHARACTER(1)        TRANSR, UPLO
```

The routine may be called by its LAPACK name ***dpftrs***.

3 Description

F07WEF (DPFTRS) is used to solve a real symmetric positive definite system of linear equations $AX = B$, the routine must be preceded by a call to F07WDF (DPFTRF) which computes the Cholesky factorization of A, where A is stored in RFP format. The solution X is computed by forward and backward substitution.

If UPLO = 'U', $A = U^{\mathrm{T}}U$, where U is upper triangular; the solution X is computed by solving $U^{\mathrm{T}}Y = B$ and then $UX = Y$.

If UPLO = 'L', $A = LL^{\mathrm{T}}$, where L is lower triangular; the solution X is computed by solving $LY = B$ and then $L^{\mathrm{T}}X = Y$.

4 References

Gustavson F G, Waśniewski J, Dongarra J J and Langou J (2010) Rectangular full packed format for Cholesky's algorithm: factorization, solution, and inversion *ACM Trans. Math. Software* **37, 2**

5 Parameters

1: TRANSR – CHARACTER(1) *Input*

On entry: specifies whether the RFP representation of A is normal or transposed.

TRANSR = 'N'
The matrix A is stored in normal RFP format.

TRANSR = 'T'
The matrix A is stored in transposed RFP format.

Constraint: TRANSR = 'N' or 'T'.

2: UPLO – CHARACTER(1) *Input*

On entry: specifies how A has been factorized.

UPLO = 'U'
$A = U^{\mathrm{T}}U$, where U is upper triangular.

UPLO = 'L'
$A = LL^{\mathrm{T}}$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

3: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

4: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides.

Constraint: NRHS ≥ 0.

5: A$(\mathrm{N} \times (\mathrm{N}+1)/2)$ – REAL (KIND=nag_wp) array *Input*

On entry: the Cholesky factorization of A stored in RFP format, as returned by F07WDF (DPFTRF).

6: B(LDB,$*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, \mathrm{NRHS})$.

On entry: the n by r right-hand side matrix B.

On exit: the n by r solution matrix X.

7: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07WEF (DPFTRS) is called.

Constraint: LDB $\geq \max(1, \mathrm{N})$.

8: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

For each right-hand side vector b, the computed solution x is the exact solution of a perturbed system of equations $(A+E)x = b$, where

if UPLO = 'U', $|E| \leq c(n)\epsilon \left|U^{\mathrm{T}}\right||U|$;

if UPLO = 'L', $|E| \leq c(n)\epsilon |L|\left|L^{\mathrm{T}}\right|$,

$c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***.

If $\hat{x}$ is the true solution, then the computed solution x satisfies a forward error bound of the form

$$\frac{\|x-\hat{x}\|_\infty}{\|x\|_\infty} \leq c(n)\,\mathrm{cond}(A,x)\epsilon$$

where $\mathrm{cond}(A,x) = \left\| |A^{-1}||A||x| \right\|_\infty / \|x\|_\infty \leq \mathrm{cond}(A) = \left\| |A^{-1}||A| \right\|_\infty \leq \kappa_\infty(A)$ and $\kappa_\infty(A)$ is the condition number when using the ∞-norm.

Note that $\mathrm{cond}(A,x)$ can be much smaller than $\mathrm{cond}(A)$.

8 Further Comments

The total number of floating point operations is approximately $2n^2r$.

The complex analogue of this routine is F07WSF (ZPFTRS).

9 Example

This example solves the system of equations $AX = B$, where

$$A = \begin{pmatrix} 4.16 & -3.12 & 0.56 & -0.10 \\ -3.12 & 5.03 & -0.83 & 1.18 \\ 0.56 & -0.83 & 0.76 & 0.34 \\ -0.10 & 1.18 & 0.34 & 1.18 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} 8.70 & 8.30 \\ -13.35 & 2.13 \\ 1.89 & 1.61 \\ -4.14 & 5.00 \end{pmatrix}.$$

Here A is symmetric positive definite, stored in RFP format, and must first be factorized by F07WDF (DPFTRF).

9.1 Program Text

```
    Program f07wefe

!     F07WEF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: dpftrf, dpftrs, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, info, ldb, lena, n, nrhs
      Character (1)                    :: transr, uplo
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:), b(:,:)
!     .. Executable Statements ..
      Write (nout,*) 'F07WEF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, nrhs, uplo, transr

      lena = n*(n+1)/2
      ldb = n
      Allocate (a(lena),b(ldb,nrhs))

!     Read A and B from data file
      Read (nin,*) a(1:lena)
      Do i = 1, n
        Read (nin,*) b(i,1:nrhs)
      End Do

!     Factorize A
      info = 0
!     The NAG name equivalent of dpftrf is f07wdf
```

```
      Call dpftrf(transr,uplo,n,a,info)

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!       Compute solution
!       The NAG name equivalent of dpftrs is f07wef
        Call dpftrs(transr,uplo,n,nrhs,a,b,ldb,info)

!       Print solution
        ifail = 0
        Call x04caf('General',' ',n,nrhs,b,ldb,'Solution(s)',ifail)

      Else
        Write (nout,*) 'A is not positive definite'
      End If

    End Program f07wefe
```

9.2 Program Data

```
F07WEF Example Program Data
  4 2 'L' 'N'                  : n, nrhs, uplo, transr
  0.76  4.16 -3.12  0.56 -0.10
  0.34  1.18  5.03 -0.83  1.18 : A
  8.70   8.30
-13.35   2.13
  1.89   1.61
 -4.14   5.00                  : matrix B
```

9.3 Program Results

```
 F07WEF Example Program Results

 Solution(s)
             1          2
 1      1.0000     4.0000
 2     -1.0000     3.0000
 3      2.0000     2.0000
 4     -3.0000     1.0000
```